Pensieve header: Verifying g1 identities for the Gaussian pairing technique.

Reminders from GWU-1612.

1-Smidgen sl_2 Let $\mathfrak{g}_1$ be the 4-dimensional Lie algebra $\mathfrak{g}_1 = \langle h, e, l, f \rangle$ over the ring $R = \mathbb{Q}[\epsilon]/(\epsilon^2 = 0)$, with h central and with $[f, l] = f$, $[e, l] = -e$, and $[e, f] = h - 2\epsilon l$.

Implementing $\mathfrak{g}_1$

```
ϵ /: ϵ^k_ /; k > 1 := 0;
PBWRule = {e → 1, l → 2, f → 3};
B[U@e, U@l] = -U@e; B[U@f, U@l] = U@f; B[U@e, U@f] = h U[] - 2 ϵ U@l;
```

```
$TD = 3; ℏ /: ℏ^d_. /; d > $TD := 0;
```

```
x_ ≤ y_ := OrderedQ[{x, y} /. PBWRule]; x_ ≺ y_ := ! OrderedQ[{y, x} /. PBWRule];
Simp[ℰ_] := Collect[ℰ, _U, Expand];
```

```
U_i_[ℰ_] := ℰ /. {h → h_i, t → t_i, u_U :→ Replace[u, x_ :→ x_i, 1]};
B[U[(x_)_i_], U[(y_)_i_]] := B[U[x_i], U[y_i]] = U_i[B[U@x, U@y]];
B[U[(x_)_i_], U[(y_)_j_]] /; i =!= j := 0;
B[x_, x_] = 0;
B[U[y_], U[x_]] := B[U[y], U[x]] = Simp[-B[U[x], U[y]]];
B[x_, y_] := x ** y - y ** x;
```

```
Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[x_] := x;
0 ** _ = _ ** 0 = 0;
x_ ** U[] := x; U[] ** x_ := x;
(a_ * x_U) ** (b_ * y_U) := If[a b === 0, 0, Simp[a b (x ** y)]];
(a_ * x_U) ** y_ := Simp[a (x ** y)]; x_ ** (a_ * y_U) := Simp[a (x ** y)];
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
```

```
U[xx___, x_] ** U[y_, yy___] := If[x ≤ y, U[xx, x, y, yy], U@xx ** (U@y ** U@x + B[U@x, U@y]) ** U@yy];
```

```
UU[l___, x_^n_, r___] := UU[l, Sequence @@ Table[x, {n}], r];
UU[l___, 1, r___] := UU[l, r];
UU[] = U[];
UU[l_, r___] := U[l] ** UU[r];
```

```
UProducts[{}, 0] = {UU[]};
UProducts[{}, n_Integer] /; n > 0 = {};
UProducts[{x_, xs___}, n_Integer] :=
  Sort@Flatten@Table[UU[x^k] ** u, {k, 0, n}, {u, UProducts[{xs}, n - k]}];
UProducts[xs_List, k_Integer, n_Integer] := UProducts[Flatten@Table[x_j, {x, xs}, {j, k}], n];
UProducts[any__, {n_}] := Flatten@Table[UProducts[any, k], {k, 0, n}];
```

```
r_i_,j_ := Simp[ℏ (h_i UU[l_j] - ϵ UU[l_i, l_j] + UU[e_i, f_j])]
```

```
UExp[u_] := Module[{s, t, k},
   s = t = U[]; k = 0;
   While[k < 20 ∧ 0 =!= (t = t ** u), s += t / (++k) !];
   Simp[s]
  ];
R_{i_,j_} := UExp[r_{i,j}];
```

```
𝕆[poly_, specs___] := Module[{vs, us, z},
  vs = Join @@ (First /@ {specs});
  us = Join @@ ({specs} /. (l_ → s_) :→ (l /. x_i_ :→ x_s));
  Simp@Total[CoefficientRules[Normal@Series[poly, {ℏ, 0, $TD}], vs] /. (p_ → c_) :→ c UU @@ (us^p)]
 ]
```

Testing YBE

```
UExp[ℏ U@e_1]
```

$$U[] + \hbar\, U[e_1] + \frac{1}{2}\hbar^2 U[e_1, e_1] + \frac{1}{6}\hbar^3 U[e_1, e_1, e_1]$$

```
R_{1,2}
```

$$U[] + \hbar h_1 U[l_2] + \left(\hbar + \frac{\hbar^2 h_1}{2} + \frac{1}{6}\hbar^3 h_1^2\right) U[e_1, f_2] - \epsilon\hbar\, U[l_1, l_2] + \frac{1}{2}\hbar^2 h_1^2 U[l_2, l_2] + \left(-\frac{\epsilon\hbar^2}{2} - \frac{1}{3}\epsilon\hbar^3 h_1\right) U[e_1, l_1, f_2] + \left(-\frac{\epsilon\hbar^2}{2} + \hbar^2 h_1 - \frac{1}{6}\epsilon\hbar^3 h_1 + \frac{1}{2}\hbar^3 h_1^2\right) U[e_1, l_2, f_2] - \epsilon\hbar^2 h_1 U[l_1, l_2, l_2] + \frac{1}{6}\hbar^3 h_1^3 U[l_2, l_2, l_2] + \left(\frac{\hbar^2}{2} - \frac{\epsilon\hbar^3}{6} + \frac{\hbar^3 h_1}{2}\right) U[e_1, e_1, f_2, f_2] + \left(-\epsilon\hbar^2 - \epsilon\hbar^3 h_1\right) U[e_1, l_1, l_2, f_2] + \left(-\frac{1}{2}\epsilon\hbar^3 h_1 + \frac{1}{2}\hbar^3 h_1^2\right) U[e_1, l_2, l_2, f_2] - \frac{1}{2}\epsilon\hbar^3 h_1^2 U[l_1, l_2, l_2, l_2] - \frac{1}{2}\epsilon\hbar^3 U[e_1, e_1, l_1, f_2, f_2] + \left(-\frac{\epsilon\hbar^3}{2} + \frac{\hbar^3 h_1}{2}\right) U[e_1, e_1, l_2, f_2, f_2] - \epsilon\hbar^3 h_1 U[e_1, l_1, l_2, l_2, f_2] + \frac{1}{6}\hbar^3 U[e_1, e_1, e_1, f_2, f_2, f_2] - \frac{1}{2}\epsilon\hbar^3 U[e_1, e_1, l_1, l_2, f_2, f_2]$$

```
$TD = 2; Simp[R_{1,2} ** R_{1,3} ** R_{2,3} - R_{2,3} ** R_{1,3} ** R_{1,2}]
```

0

```
$TD = 3; Simp[R_{1,2} ** R_{1,3} ** R_{2,3} - R_{2,3} ** R_{1,3} ** R_{1,2}]
```

0

```
$TD = 4; Simp[R_{1,2} ** R_{1,3} ** R_{2,3} - R_{2,3} ** R_{1,3} ** R_{1,2}]
```

$$\left(-\frac{1}{12}\epsilon\hbar^4 h_1 h_2 - \frac{1}{12}\epsilon\hbar^4 h_2^2\right) U[e_1, l_2, f_3] + \left(\frac{1}{12}\epsilon\hbar^4 h_1 h_2 - \frac{1}{12}\epsilon\hbar^4 h_2^2\right) U[e_1, l_3, f_3] - \frac{1}{6}\epsilon\hbar^4 h_2 U[e_1, e_1, f_2, f_3] + \frac{1}{6}\epsilon\hbar^4 h_2 U[e_1, e_1, f_3, f_3] + \left(\frac{1}{12}\epsilon\hbar^4 h_1 - \frac{1}{12}\epsilon\hbar^4 h_2\right) U[e_1, e_2, f_2, f_3] + \left(-\frac{1}{12}\epsilon\hbar^4 h_1 - \frac{1}{12}\epsilon\hbar^4 h_2\right) U[e_1, e_2, f_3, f_3]$$

$\mathfrak{g}_0$ tests

```
$TD = 6; 𝕆[Exp[ℏ h_1 l_2 + (ⅇ^{ℏ h_1} - 1)/h_1 e_1 f_2], {e_1} → 1, {l_2, f_2} → 2] == R_{1,2} /. ϵ → 0
```

True

```
$TD = 3; 𝕆[ⅇ^{ℏ δ e_1 f_1}, {f_1, e_1} → 1] /. ϵ → 0
```

$$\left(1 - \delta\hbar h_1 + \delta^2\hbar^2 h_1^2 - \delta^3\hbar^3 h_1^3\right) U[] + \left(\delta\hbar - 2\delta^2\hbar^2 h_1 + 3\delta^3\hbar^3 h_1^2\right) U[e_1, f_1] + \left(\frac{\delta^2\hbar^2}{2} - \frac{3}{2}\delta^3\hbar^3 h_1\right) U[e_1, e_1, f_1, f_1] + \frac{1}{6}\delta^3\hbar^3 U[e_1, e_1, e_1, f_1, f_1, f_1]$$

```
$TD = 3; With[{ν = (1 + ħ h1 δ)^-1}, 𝕆[ν e^(ħ ν δ e1 f1), {e1, f1} → 1]] /. ϵ → 0
```

$$\left(1-\delta\hbar h_1+\delta^2\hbar^2 h_1^2-\delta^3\hbar^3 h_1^3\right)\mathrm{U}[\,]+\left(\delta\hbar-2\delta^2\hbar^2 h_1+3\delta^3\hbar^3 h_1^2\right)\mathrm{U}[e_1,f_1]+\left(\frac{\delta^2\hbar^2}{2}-\frac{3}{2}\delta^3\hbar^3 h_1\right)\mathrm{U}[e_1,e_1,f_1,f_1]+\frac{1}{6}\delta^3\hbar^3\,\mathrm{U}[e_1,e_1,e_1,f_1,f_1,f_1]$$

```
$TD = 6; With[{ν = (1 + ħ h1 δ)^-1}, 𝕆[e^(ħ δ e1 f1), {f1, e1} → 1] == 𝕆[ν e^(ħ ν δ e1 f1), {e1, f1} → 1] /. ϵ → 0]
```

True

$\mathfrak{g}_1$ tests

```
Λ[ν_] := Expand[
  1/(2 μ^4) (-b α^2 β^2 + u^2 β^2 δ (2 + b δ) + w^2 δ (α + u δ) (α (2 + b δ) + u δ (4 + 3 b δ)) -
      4 b α β δ μ + 4 c α β μ^2 - 2 b δ^2 μ^2 + 4 c δ μ^3 + 2 u β (α β + 2 δ μ (1 + c μ)) +
      2 w (α^2 β + 2 α δ (u β (2 + b δ) + μ (1 + c μ)) + u δ^2 (u β (3 + 2 b δ) + 2 μ (2 + b δ + c μ)))) /.
   {α | β → 0, μ → ν^-1, b → ħ h1, c → l1, u → ħ e1, w → f1}
 ]
```

```
$TD = 6; With[
 {ν = (1 + ħ h1 δ)^-1},
 Simp[𝕆[e^(ħ δ e1 f1), {f1, e1} → 1] - 𝕆[ν (1 + ϵ ħ Λ[ν]) e^(ħ ν δ e1 f1), {e1, l1, f1} → 1]]
]
```

0

```
Λ[ν] /. {ħ → 1, x_1 :> x}
```

$$2\,l\,\delta\,\nu-h\,\delta^2\nu^2+2\,e\,f\,l\,\delta^2\nu^2+4\,e\,f\,\delta^2\nu^3+2\,e\,f\,h\,\delta^3\nu^3+2\,e^2f^2\delta^3\nu^4+\frac{3}{2}e^2f^2h\,\delta^4\nu^4$$

```
TeXForm[Λ[ν] /. {ħ → 1, x_1 :> x}]
```

```
2 \delta ^3 e^2 f^2 \nu ^4+\frac{3}{2} \delta ^4 e^2 f^2 h \nu ^4+4 \delta ^2 e f \nu ^3+2 \delta ^3 e f h
   ^2 h \nu ^2+2 \delta  l \nu
```

Computations for Gaussian Pairing

```
Simplify[1/(2 μ^3) (-b α^2 β^2 + u^2 β^2 δ (2 + b δ) + w^2 δ (α + u δ) (α (2 + b δ) + u δ (4 + 3 b δ)) -
     4 b α β δ μ + 4 c α β μ^2 - 2 b δ^2 μ^2 + 4 c δ μ^3 + 2 u β (α β + 2 δ μ (1 + c μ)) +
     2 w (α^2 β + 2 α δ (u β (2 + b δ) + μ (1 + c μ)) + u δ^2 (u β (3 + 2 b δ) + 2 μ (2 + b δ + c μ)))) /. {δ → 0, μ → 1}]
```

$$\frac{1}{2}\alpha\beta\left(4c+2w\alpha+2u\beta-b\alpha\beta\right)$$

```
Simplify[1/(2 μ^4) (-b α^2 β^2 + u^2 β^2 δ (2 + b δ) + w^2 δ (α + u δ) (α (2 + b δ) + u δ (4 + 3 b δ)) -
     4 b α β δ μ + 4 c α β μ^2 - 2 b δ^2 μ^2 + 4 c δ μ^3 + 2 u β (α β + 2 δ μ (1 + c μ)) +
     2 w (α^2 β + 2 α δ (u β (2 + b δ) + μ (1 + c μ)) + u δ^2 (u β (3 + 2 b δ) + 2 μ (2 + b δ + c μ)))) /.
  {δ → 0, μ → 1, b → ħ h1, c → l1, u → ħ e1, w → f1}]
```

$$\frac{1}{2}\alpha\beta\left(2\beta\hbar e_1+2\alpha f_1-\alpha\beta\hbar h_1+4\,l_1\right)$$

```
$TD = 8; Simp[𝕆[e^(ħ (α f1 + β e1)), {f1, e1} → 1] -
   𝕆[(1 - 1/2 α β ϵ ħ^2 (-2 β ħ e1 - 2 α ħ f1 + α β ħ^2 h1 - 4 l1)) e^(ħ (-ħ h1 α β + α f1 + β e1)), {e1, l1, f1} → 1]]
```

0

```
Simplify[
 1/(2 μ^4) (-b α^2 β^2 + u^2 β^2 δ (2 + b δ) + w^2 δ (α + u δ) (α (2 + b δ) + u δ (4 + 3 b δ)) - 4 b α β δ μ + 4 c α β μ^2 - 2 b δ^2 μ^2 +
    4 c δ μ^3 + 2 u β (α β + 2 δ μ (1 + c μ)) + 2 w (α^2 β + 2 α δ (u β (2 + b δ) + μ (1 + c μ)) +
       u δ^2 (u β (3 + 2 b δ) + 2 μ (2 + b δ + c μ)))) /. {b → h, c → l, u → e, w → f}]
```

$$\frac{1}{2\mu^4}\left(-h\alpha^2\beta^2+e^2\beta^2\delta\left(2+h\delta\right)+f^2\delta\left(\alpha+e\delta\right)\left(\alpha\left(2+h\delta\right)+e\delta\left(4+3h\delta\right)\right)-4h\alpha\beta\delta\mu+4l\alpha\beta\mu^2-2h\delta^2\mu^2+4l\delta\mu^3+2e\beta\left(\alpha\beta+2\delta\mu\left(1+l\mu\right)\right)+2f\left(\alpha^2\beta+2\alpha\delta\left(e\beta\left(2+h\delta\right)+\mu\left(1+l\mu\right)\right)+e\delta^2\left(e\beta\left(3+2h\delta\right)+2\mu\left(2+h\delta+l\mu\right)\right)\right)\right)$$

```
Simplify[
 1/(2 μ^4) (-b α^2 β^2 + u^2 β^2 δ (2 + b δ) + w^2 δ (α + u δ) (α (2 + b δ) + u δ (4 + 3 b δ)) - 4 b α β δ μ + 4 c α β μ^2 - 2 b δ^2 μ^2 +
    4 c δ μ^3 + 2 u β (α β + 2 δ μ (1 + c μ)) + 2 w (α^2 β + 2 α δ (u β (2 + b δ) + μ (1 + c μ)) +
       u δ^2 (u β (3 + 2 b δ) + 2 μ (2 + b δ + c μ)))) /. {b → h, c → l, u → e, w → f, α → 0, β → 0}]
```

$$\frac{1}{2\mu^4}\delta\left(e^2f^2\delta^2\left(4+3h\delta\right)+4ef\delta\mu\left(2+h\delta+l\mu\right)+2\mu^2\left(-h\delta+2l\mu\right)\right)$$